

SAS[®] 9 Graph Style Tip Sheet

Using Style Elements

You can use style elements to control attributes of plot features such as markers and lines as follows:

```
proc sgplot data=sashelp.class noautolegend;
  pbspline y=weight x=height /
  markerattrs=GraphData2
  lineattrs=GraphFit2;
run;
```

You can set most attributes using style elements and override certain attributes with options as follows:

```
proc sgplot data=sashelp.class noautolegend;
  pbspline y=weight x=height /
  markerattrs=Graphdata2(symbol=circlefilled)
  lineattrs=GraphFit2(thickness=1px);
run;
```

In the GTL, you can specify both style elements and specific attributes as follows:

```
proc template;
  define statgraph pbs;
    begingraph;
      layout overlay;
        scatterplot y=weight x=height /
        markerattrs=GraphData5
        (color=GraphData2:ContrastColor);
        pbsplineplot y=weight x=height /
        lineattrs=GraphData8
        (color=GraphFit2:Color);
      endlayout;
    endgraph;
  end;
run;
```

```
proc sgrender data=sashelp.class template=pbs;
run;
```

The marker style comes from GraphData5 but with the GraphData2 contrast color. The line style comes from GraphData8 but with the GraphFit2 color.


Style Template Creation Example

```
proc template;
  define style styles.MyStyle;
    parent=styles.default;

    class GraphDataDefault /
      Color = cxCAD5E5
      ContrastColor = cx003399
      LineStyle = 1
      LineThickness = 1px
      MarkerSymbol = "StarFilled"
      MarkerSize = 5px
      StartColor = cxAFB5A6
      NeutralColor = cxFFFFFF
      EndColor = cx667FA2;

    style colors from colors /
      'docbg' = cxFAFBFE;
  end;
run;

ods html style=mystyle;
ods graphics on / width=3in;
proc sgplot data=sashelp.class noautolegend;
  title 'Polynomial Fit';
  reg y=weight x=height / degree=3;
run;
ods html close;
```



For more information, see:

Kuhfeld, W. F. 2010. *Statistical Graphics in SAS[®]: An Introduction to the Graph Template Language and the Statistical Graphics Procedures*. Cary, NC: SAS Press.
support.sas.com/publishing/authors/kuhfeld.html

For complete information, see the SAS[®] 9.2 documentation at
<http://support.sas.com/v9doc>

SAS[®] 9

Graph Style Tip Sheet

This tip sheet places frequently used information in one place, on one sheet of paper, so you don't have to search through the online documentation. It also gives you something to take home, type in, and try.

ODS Graphics is an extension of ODS (the Output Delivery System). ODS styles control the colors and general appearance of all graphs and tables, and SAS provides several styles that are recommended for use with statistical graphics. The default style that you see when you run SAS depends on the ODS destination, system options, and SAS registry settings.

This tip sheet presents an overview of ODS styles and style elements and their use in ODS Graphics in SAS 9.2.



SAS Institute Inc. World Headquarters
+1 919 677 8000 www.sas.com/offices

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies. Copyright © 2011, SAS Institute Inc. All rights reserved. 295422_0211


SAS[®]9 Graph Style Tip Sheet

ODS Styles

Style (Default in Destination or Application)


DEFAULT (HTML)

Parameter Estimates					
Variable	DF	Parameter Estimate	Standard Error	t Value	Pr > t
Intercept	1	-143.02692	32.27459	-4.43	0.0004
Height	1	3.89903	0.51609	7.55	<.0001




LISTING (LISTING)

Parameter Estimates					
Variable	DF	Parameter Estimate	Standard Error	t Value	Pr > t
Intercept	1	-143.02692	32.27459	-4.43	0.0004
Height	1	3.89903	0.51609	7.55	<.0001




RTF (RTF)

Parameter Estimates					
Variable	DF	Parameter Estimate	Standard Error	t Value	Pr > t
Intercept	1	-143.02692	32.27459	-4.43	0.0004
Height	1	3.89903	0.51609	7.55	<.0001




STATISTICAL (SAS/STAT documentation)

Parameter Estimates					
Variable	DF	Parameter Estimate	Standard Error	t Value	Pr > t
Intercept	1	-143.02692	32.27459	-4.43	0.0004
Height	1	3.89903	0.51609	7.55	<.0001




ANALYSIS (Enterprise Guide)

Parameter Estimates					
Variable	DF	Parameter Estimate	Standard Error	t Value	Pr > t
Intercept	1	-143.02692	32.27459	-4.43	0.0004
Height	1	3.89903	0.51609	7.55	<.0001



JOURNAL, JOURNAL2

Parameter Estimates					
Variable	DF	Parameter Estimate	Standard Error	t Value	Pr > t
Intercept	1	-143.02692	32.27459	-4.43	0.0004
Height	1	3.89903	0.51609	7.55	<.0001



Displaying Styles

Use PROC TEMPLATE to display style definitions:

```
proc template;
  list styles;           * list all styles;
  source styles.statistical; *display STATISTICAL;
  source styles.default; *display DEFAULT style;
run;
```

Many styles (including STATISTICAL, LISTING, ANALYSIS, and JOURNAL) inherit from the DEFAULT style. You need display the style and all of its parents to see the complete style definition.

Line Styles, Marker Names, Colors

LineStyle name (used in GTL and style) and number (used in style definitions): Dash 20, DashDashDot 14, DashDotDot 15, Dot 34, LongDash 5, LongDashShortDash 26, MediumDash 4, MediumDashDotDot 42, MediumDashShortDash 8, ShortDash 2, ShortDashDot 41, Solid 1, ThinDot 35.

MarkerSymbol values: ArrowDown Asterisk Circle CircleFilled Diamond DiamondFilled GreaterThan Hash HomeDown HomeDownFilled IBeam Plus Square SquareFilled Star StarFilled Tack Tilde Triangle TriangleFilled Union X Y Z

Colors: Color applies to filled areas, and contrast color applies to markers and lines. Colors can be specified in values of the form CXrrggbb, where the last six characters specify RGB (red, green, blue) values on the hexadecimal scale of 00 to FF (0 to 255 base 10). Common colors are CXFF0000 (red), CX00FF00 (green), CX0000FF (blue), CXFFFF00 (yellow, a mix of red and green), CXFF00FF (magenta, a mix of red and blue), CX00FFFF (cyan, a mix of green and blue), CXFFFFFF (white, a mix of red, green, and blue), CX000000 (black, no color), CXDDDDDD (very light gray), CX222222 (very dark gray), and so on. Common color names such as red, green, blue, and so on can also be used.

Graph Style Elements

You refer to ODS style elements in graph templates with *style-element* or *style-element:attribute* (for example, **GraphDataDefault:ContrastColor**). Common graph style elements include:

Graph graph size, outer border appearance
GraphConfidence primary fit confidence interval
GraphData1 first grouped data item attributes
GraphData12 twelfth grouped data item attributes
GraphDataDefault not grouped data item attributes
GraphFit primary fit function
GraphFit2 secondary fit function
GraphGridLines horizontal and vertical grid lines
GraphOutlier outlier data attributes
GraphPredictionLimits fills for prediction limits
GraphReference reference lines and drop lines
GraphDataText attributes of point and line labels
GraphValueText attributes of axis tick values
GraphLabelText attrs of axis labels and legend title
GraphFootnoteText footnotes
GraphTitleText titles
GraphWalls vertical walls bounded by axes

Common attributes and sample values include:

```
BackgroundColor = colors("docbg")
Color = GraphColors("gdata")
ContrastColor = GraphColors("gdata")
Displayopts = "fill outline"
EndColor = GraphColors("gconramp3cend")
Font = Fonts("TitleFont")
FrameBorder = on
LineStyle = 1
LineThickness = 3px
MarkerSize = 7px
MarkerSymbol = "circle"
NeutralColor = GraphColors("gconramp3neutral")
Padding = 7
StartColor = GraphColors("gconramp3start")
```

Many attributes are defined indirectly by default. Not all attributes can be used with all elements.

DEFAULT Style Graph Fonts

The graph fonts for the DEFAULT style are:

```
class GraphFonts
  "Fonts used in graph styles" /
  'GraphDataFont' = ("

```

You can use PROC TEMPLATE with the SOURCE statement to display other styles and see other font definitions. You can create a new style with modified font definitions as follows:

```
proc template;
  define style myfonts;
    parent = styles.default;
    style GraphFonts from GraphFonts /
      'GraphDataFont' = ("Arial",7pt)
      'GraphLabelFont' = ("Times New Roman",
      10pt)
      'GraphFootnoteFont' = ("Courier New",
      10pt);
  end;
run;
```

The style MYSTYLE differs from the DEFAULT style only in the three font definitions. Specify STYLE=MYSTYLE on an ODS destination statement to use your new style.